

Nombre y Apellidos:
Especialidad y Grupo:

FUNDAMENTOS DE INFORMÁTICA

10-septiembre-2001

PROGRAMA de Estructuras alternativas y repetitivas (2 puntos)

1. Realizar la operación aritmética de dos operandos. Para ello:

- Teclar los números reales **a** y **b**, correspondientes a los dos operandos. Si el valor de **b** es igual a 0, entonces emitir el siguiente mensaje:

*** El segundo operando no puede ser cero. ***

y terminar el programa.

- En caso contrario,
 - Teclar un carácter **ope** correspondiente a la operación aritmética. Controlar que el operador aritmético sea +, -, * o /, repitiendo su entrada hasta que sea correcto.
 - Mostrar por pantalla el resultado de la siguiente forma:
 - Por ejemplo, si los operandos son 9.9 y -3 y el operador es la división, entonces:

9.9 / -3 = -3.3

```
REM Programa para realizar una operación aritmética.  
REM
```

```
DIM a AS SINGLE  
DIM b AS SINGLE  
DIM ope AS STRING * 1  
DIM rdo AS SINGLE  
CLS
```

```
INPUT "Teclear el operando a: ", a  
INPUT "Teclear el operando b: ", b  
IF b = 0 THEN  
    PRINT "*** El segundo operando no puede ser cero. ***"  
ELSE  
    DO  
        INPUT "Teclear el operador (+,-,* o /): ", ope  
    LOOP UNTIL ope = "+" OR ope = "-" OR ope = "*" OR ope = "/"  
    SELECT CASE ope  
        CASE IS = "+"  
            rdo = a + b  
        CASE IS = "-"  
            rdo = a - b  
        CASE IS = "*"  
            rdo = a * b  
        CASE IS = "/"  
            rdo = a / b  
    END SELECT  
    PRINT a; ope; b; " = "; rdo  
END IF
```

PROGRAMA de Arrays y Registros (3 puntos)

2. La pizzería “PizzaSurprise” se caracteriza por que no se pueden elegir los ingredientes de la pizza. Dichos componentes son obtenidos al azar por el programa en QB mediante el uso de las funciones RANDOMIZE y RND.

Nuestra pizzería dispone de 20 ingredientes diferentes. Para realizar el programa se necesitan los siguientes arrays.

- Lista alfanumérica, llamada *ingredientesPizzeria*, que contiene los 20 ingredientes con los que se puede confeccionar una pizza.
- Lista dinámica alfanumérica, denominada *ingredientesNuestraPizza*, que contiene los ingredientes aleatorios de nuestra pizza.

Se pide:

1. Cargar la lista *ingredientesPizzeria* como queráis.
2. Cargar la lista *ingredientesNuestraPizza* de la siguiente forma:
 - Preguntar al cliente:
¿Cuántos ingredientes quieres? _
 - Dimensionar ahora la lista dinámica teniendo en cuenta la respuesta anterior.
 - Para la obtención de cada ingrediente se sigue este proceso:
 - El ordenador genera un número aleatorio comprendido entre 1 y 20. Dicho número representa el índice para acceder al array unidimensional *ingredientesPizzeria*.
 - Por ejemplo, si dicho número es el 3, entonces en la primera posición de *ingredientesNuestraPizza* se debe almacenar el ingrediente Tomate.

<i>ingredientesPizzeria</i>	
1	Bonito
2	Queso
3	Tomate
...	
20	Aceitunas

3. Ordenar alfabéticamente la lista *ingredientesNuestraPizza*.
4. Visualizar por pantalla la factura de nuestra pizza. Dado que los componentes para la confección de nuestra pizza pueden aparecer una sola vez, por duplicado, por triplicado, etc., cada ingrediente diferente se debe visualizar en una línea diferente según este formato:

Pizzería PIZZASURPRISE

Aceitunas
Bonito Bonito
Cebolla Cebolla Cebolla

.....

ingredientes x 150 pesetas = ##,### pesetas.
ingredientes x 0.90 euros = ###.## euros.

```

REM Visualizar por pantalla los ingredientes de una pizza
REM y la factura correspondiente a dicha pizza
REM pedida en PIZZASURPRISE.
REM

DIM ingredientesPizzeria(1 TO 20) AS STRING
DIM i AS INTEGER, j AS INTEGER
RANDOMIZE TIMER
CLS

'Carga de la lista ingredientesPizzeria
DATA Lechuga, Tomate, Cebolla, Maíz, Zanahoria
DATA Pimiento, Aceituna, Pepinillo, Espárrago, Guisante
DATA Bonito, Anchoa, Atún, Huevo, Chaca
DATA Mortadela, Chorizo, Jamón york, Jamón serrano, Salchichón

FOR i = 1 TO 20
    READ ingredientesPizzeria(i)
NEXT i

DIM n AS INTEGER
PRINT " ¿Cuántos ingredientes quieres";
INPUT n

DIM ingredientesNuestraPizza(1 TO n) AS STRING
DIM num AS INTEGER

'Carga de la lista ingredientesNuestraPizza
FOR i = 1 TO n
    num = INT(RND * 20) + 1
    ingredientesNuestraPizza(i) = ingredientesPizzeria(num)
NEXT i

'Ordenación alfabética de la lista ingredientesNuestraPizza
FOR i = 1 TO n - 1
    FOR j = 1 TO n - i
        IF ingredientesNuestraPizza(j) > ingredientesNuestraPizza(j +
1) THEN
            SWAP ingredientesNuestraPizza(j),
ingredientesNuestraPizza(j + 1)
        END IF
    NEXT j
NEXT i

'Visualización ordenada de los ingredientes
FOR i = 1 TO n
    PRINT ingredientesNuestraPizza(i)
NEXT i

'Emisión de la factura correspondiente a una pizza
PRINT
PRINT "Pizzería PIZZASURPRISE"
PRINT STRING$(22, "-")
FOR i = 1 TO n
    IF i = 1 THEN
        PRINT SPC(5);
        PRINT ingredientesNuestraPizza(i);
    ELSE
        IF ingredientesNuestraPizza(i) <> ingredientesNuestraPizza(i -
1) THEN
            PRINT
            END IF
        PRINT SPC(5);
        PRINT ingredientesNuestraPizza(i);
    END IF
NEXT i

```

```
NEXT i
PRINT
PRINT STRING$(50, "-")
PRINT USING "##"; n;
PRINT " ingredientes x 150 pesetas = ";
PRINT USING "##,###"; n * 150;
PRINT "      pesetas."
PRINT USING "##"; n;
PRINT " ingredientes x 0.90 euros   = ";
PRINT USING "   ###.##"; n * .9;
PRINT " euros."
```

CUESTIÓN de Funciones (2.5 puntos)

3. Utilizando las funciones intrínsecas o propias de QB, codificar para cada apartado un programa de tal forma que si el dato de entrada es ... la salida por pantalla sea ...
- Entrada =
AMOR
Salida =
R
O
M
A
 - Generalizar el programa anterior para que sea independiente de la palabra introducida.
 - Entrada =
123.4567
Salida =
4567
123
 - Generalizar el programa anterior con el fin de que sea válido para cualquier número real positivo.

```
REM Programa para la utilización de las funciones de QB.  
REM
```

```
CLS
```

```
'Apartado a  
DIM palabra1 AS STRING * 4  
DIM i AS INTEGER  
palabra1 = "AMOR"  
FOR i = 4 TO 1 STEP -1  
    PRINT MID$(palabra1, i, 1)  
NEXT i
```

```
'Apartado b  
DIM palabra2 AS STRING  
DIM j AS INTEGER  
INPUT "Teclear una palabra: ", palabra2  
FOR j = LEN(palabra2) TO 1 STEP -1  
    PRINT MID$(palabra2, j, 1)  
NEXT j
```

```
'Apartado c  
DIM real1 AS SINGLE  
DIM cadenaReal1 AS STRING  
real1 = 123.4567  
cadenaReal1 = LTRIM$(STR$(real1))  
PRINT MID$(cadenaReal1, 5, 4)  
PRINT MID$(cadenaReal1, 1, 3)
```

```
'Apartado d  
DIM real2 AS DOUBLE  
DIM cadenaReal2 AS STRING  
DIM punto2 AS INTEGER  
INPUT "Teclear un número real positivo: ", real2  
cadenaReal2 = LTRIM$(STR$(real2))  
punto2 = INSTR(cadenaReal2, ".")  
PRINT MID$(cadenaReal2, punto2 + 1, LEN(cadenaReal2)-punto2)  
PRINT MID$(cadenaReal2, 1, punto2 - 1)
```

PROGRAMA de Procedimientos (2.5 puntos)

4. Visualizar por pantalla el acrónimo de un organismo. Para ello:

- Programa Principal
 - Teclear el nombre del organismo. Se supone que el primer carácter de cada palabra se introduce de forma correcta, es decir, en mayúsculas.
 - Utilizar un procedimiento llamado acronimoOrga.
 - Visualizar en pantalla en una línea el nombre del organismo y su acrónimo.
 - Por ejemplo:

Comunidad Económica Europea	C.E.E.
Estados Unidos	EE.UU.
- Procedimiento acronimoOrga
 - El parámetro formal de entrada es el nombre del organismo y el de salida es su acrónimo.
 - Para obtener el acrónimo de forma adecuada también hay que tener en cuenta aquellos nombres de organismos en plural (que acaban en “s”), pues en este caso la inicial aparece duplicada.

```
REM Visualizar por pantalla el acrónimo de un organismo.
REM

DECLARE SUB acronimoOrga (nombreOrg AS STRING, acroOrg AS STRING)

CLS

DIM nombre AS STRING
DIM acronimo AS STRING

PRINT "Teclear nombre organismo: ", nombre

CALL acronimoOrga(nombre, acronimo)

PRINT
PRINT nombre;
PRINT SPC(5);
COLOR 10
PRINT acronimo
END

SUB acronimoOrga (nombreOrg AS STRING, acroOrg AS STRING)
  DIM i AS INTEGER, j AS INTEGER
  DIM letra AS STRING * 1
  DIM palabra AS STRING
  FOR i = 1 TO LEN(nombreOrg)
    letra = MID$(nombreOrg, i, 1)
    IF letra <> " " THEN
      palabra = palabra + letra
      j = j + 1
    END IF
    IF letra = " " OR i = LEN(nombreOrg) THEN
      acroOrg = acroOrg + UCASE$(MID$(palabra, 1, 1))
      IF MID$(palabra, j, 1) = "s" THEN
        acroOrg = acroOrg + UCASE$(MID$(palabra, 1, 1))
      END IF
      acroOrg = acroOrg + "."
      palabra = ""
      j = 0
    END IF
  NEXT i
END SUB
```

